// File: @openzeppelin/contracts/access/Ownable.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts (last updated v4.7.0) (access/Ownable.sol)

pragma solidity ^0.8.0;

import "../utils/Context.sol";

/**
 * @dev Contract module which provides a basic access control mechanism, where
 * there is an account (an owner) that can be granted exclusive access to
 * specific functions.
 *
 * By default, the owner account will be the one that deploys the contract. This
 * can later be changed with {transferOwnership}.
 *
 * This module is used through inheritance. It will make available the modifier
 * `onlyOwner`, which can be applied to your functions to restrict their use to
 * the owner.
 */
abstract contract Ownable is Context {
    address private _owner;

    event OwnershipTransferred(address indexed previousOwner, address indexed newOwner);

    /**
     * @dev Initializes the contract setting the deployer as the initial owner.
     */
    constructor() {
        _transferOwnership(_msgSender());
    }

    /**
     * @dev Throws if called by any account other than the owner.
     */
    modifier onlyOwner() {
        _checkOwner();
        _;
    }

    /**
     * @dev Returns the address of the current owner.
     */
    function owner() public view virtual returns (address) {
        return _owner;
    }

    /**
     * @dev Throws if the sender is not the owner.
     */
    function _checkOwner() internal view virtual {
        require(owner() == _msgSender(), "Ownable: caller is not the owner");
    }

    /**
     * @dev Leaves the contract without owner. It will not be possible to call
     * `onlyOwner` functions anymore. Can only be called by the current owner.
     *
     * NOTE: Renouncing ownership will leave the contract without an owner,
     * thereby removing any functionality that is only available to the owner.
     */
    function renounceOwnership() public virtual onlyOwner {
        _transferOwnership(address(0));
    }

    /**
     * @dev Transfers ownership of the contract to a new account (`newOwner`).
     * Can only be called by the current owner.
     */
    function transferOwnership(address newOwner) public virtual onlyOwner {
        require(newOwner != address(0), "Ownable: new owner is the zero address");
        _transferOwnership(newOwner);
    }

    /**
     * @dev Transfers ownership of the contract to a new account (`newOwner`).
     * Internal function without access restriction.
     */
    function _transferOwnership(address newOwner) internal virtual {
        address oldOwner = _owner;
        _owner = newOwner;
        emit OwnershipTransferred(oldOwner, newOwner);
    }
}

// File: @openzeppelin/contracts/finance/PaymentSplitter.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts (last updated v4.8.0) (finance/PaymentSplitter.sol)

pragma solidity ^0.8.0;

import "../token/ERC20/utils/SafeERC20.sol";
import "../utils/Address.sol";
import "../utils/Context.sol";

/**
 * @title PaymentSplitter
 * @dev This contract allows to split Ether payments among a group of accounts. The sender does not need to be aware
 * that the Ether will be split in this way, since it is handled transparently by the contract.
 *
 * The split can be in equal parts or in any other arbitrary proportion. The way this is specified is by assigning each
 * account to a number of shares. Of all the Ether that this contract receives, each account will then be able to claim
 * an amount proportional to the percentage of total shares they were assigned. The distribution of shares is set at the
 * time of contract deployment and can't be updated thereafter.
 *
 * `PaymentSplitter` follows a _pull payment_ model. This means that payments are not automatically forwarded to the
 * accounts but kept in this contract, and the actual transfer is triggered as a separate step by calling the {release}
 * function.
 *
 * NOTE: This contract assumes that ERC20 tokens will behave similarly to native tokens (Ether). Rebasing tokens, and
 * tokens that apply fees during transfers, are likely to not be supported as expected. If in doubt, we encourage you
 * to run tests before sending real value to this contract.
 */
contract PaymentSplitter is Context {
    event PayeeAdded(address account, uint256 shares);
    event PaymentReleased(address to, uint256 amount);
    event ERC20PaymentReleased(IERC20 indexed token, address to, uint256 amount);
    event PaymentReceived(address from, uint256 amount);

    uint256 private _totalShares;
    uint256 private _totalReleased;

    mapping(address => uint256) private _shares;
    mapping(address => uint256) private _released;
    address[] private _payees;

    mapping(IERC20 => uint256) private _erc20TotalReleased;
    mapping(IERC20 => mapping(address => uint256)) private _erc20Released;

    /**
     * @dev Creates an instance of `PaymentSplitter` where each account in `payees` is assigned the number of shares at
     * the matching position in the `shares` array.
     *
     * All addresses in `payees` must be non-zero. Both arrays must have the same non-zero length, and there must be no
     * duplicates in `payees`.
     */
    constructor(address[] memory payees, uint256[] memory shares_) payable {
        require(payees.length == shares_.length, "PaymentSplitter: payees and shares length mismatch");
        require(payees.length > 0, "PaymentSplitter: no payees");

        for (uint256 i = 0; i < payees.length; i++) {
            _addPayee(payees[i], shares_[i]);
        }
    }

    /**
     * @dev The Ether received will be logged with {PaymentReceived} events. Note that these events are not fully
     * reliable: it's possible for a contract to receive Ether without triggering this function. This only affects the
     * reliability of the events, and not the actual splitting of Ether.
     *
     * To learn more about this see the Solidity documentation for
     * https://solidity.readthedocs.io/en/latest/contracts.html#fallback-function[fallback
     * functions].
     */
    receive() external payable virtual {
        emit PaymentReceived(_msgSender(), msg.value);
    }

    /**
     * @dev Getter for the total shares held by payees.
     */
    function totalShares() public view returns (uint256) {
        return _totalShares;
    }

    /**
     * @dev Getter for the total amount of Ether already released.
     */
    function totalReleased() public view returns (uint256) {
        return _totalReleased;
    }

    /**
     * @dev Getter for the total amount of `token` already released. `token` should be the address of an IERC20
     * contract.
     */
    function totalReleased(IERC20 token) public view returns (uint256) {
        return _erc20TotalReleased[token];
    }

    /**
     * @dev Getter for the amount of shares held by an account.
     */
    function shares(address account) public view returns (uint256) {
        return _shares[account];
    }

    /**
     * @dev Getter for the amount of Ether already released to a payee.
     */
    function released(address account) public view returns (uint256) {
        return _released[account];
    }

    /**
     * @dev Getter for the amount of `token` tokens already released to a payee. `token` should be the address of an
     * IERC20 contract.
     */
    function released(IERC20 token, address account) public view returns (uint256) {
        return _erc20Released[token][account];
    }

    /**
     * @dev Getter for the address of the payee number `index`.
     */
    function payee(uint256 index) public view returns (address) {
        return _payees[index];
    }

    /**
     * @dev Getter for the amount of payee's releasable Ether.
     */
    function releasable(address account) public view returns (uint256) {
        uint256 totalReceived = address(this).balance + totalReleased();
        return _pendingPayment(account, totalReceived, released(account));
    }

    /**
     * @dev Getter for the amount of payee's releasable `token` tokens. `token` should be the address of an
     * IERC20 contract.
     */
    function releasable(IERC20 token, address account) public view returns (uint256) {
        uint256 totalReceived = token.balanceOf(address(this)) + totalReleased(token);
        return _pendingPayment(account, totalReceived, released(token, account));
    }

    /**
     * @dev Triggers a transfer to `account` of the amount of Ether they are owed, according to their percentage of the
     * total shares and their previous withdrawals.
     */
    function release(address payable account) public virtual {
        require(_shares[account] > 0, "PaymentSplitter: account has no shares");

        uint256 payment = releasable(account);

        require(payment != 0, "PaymentSplitter: account is not due payment");

        // _totalReleased is the sum of all values in _released.
        // If "_totalReleased += payment" does not overflow, then "_released[account] += payment" cannot overflow.
        _totalReleased += payment;
        unchecked {
            _released[account] += payment;
        }

        Address.sendValue(account, payment);
        emit PaymentReleased(account, payment);
    }

    /**
     * @dev Triggers a transfer to `account` of the amount of `token` tokens they are owed, according to their
     * percentage of the total shares and their previous withdrawals. `token` must be the address of an IERC20
     * contract.
     */
    function release(IERC20 token, address account) public virtual {
        require(_shares[account] > 0, "PaymentSplitter: account has no shares");

        uint256 payment = releasable(token, account);

        require(payment != 0, "PaymentSplitter: account is not due payment");

        // _erc20TotalReleased[token] is the sum of all values in _erc20Released[token].
        // If "_erc20TotalReleased[token] += payment" does not overflow, then "_erc20Released[token][account] += payment"
        // cannot overflow.
        _erc20TotalReleased[token] += payment;
        unchecked {
            _erc20Released[token][account] += payment;
        }

        SafeERC20.safeTransfer(token, account, payment);
        emit ERC20PaymentReleased(token, account, payment);
    }

    /**
     * @dev internal logic for computing the pending payment of an `account` given the token historical balances and
     * already released amounts.
     */
    function _pendingPayment(
        address account,
        uint256 totalReceived,
        uint256 alreadyReleased
    ) private view returns (uint256) {
        return (totalReceived * _shares[account]) / _totalShares - alreadyReleased;
    }

    /**
     * @dev Add a new payee to the contract.
     * @param account The address of the payee to add.
     * @param shares_ The number of shares owned by the payee.
     */
    function _addPayee(address account, uint256 shares_) private {
        require(account != address(0), "PaymentSplitter: account is the zero address");
        require(shares_ > 0, "PaymentSplitter: shares are 0");
        require(_shares[account] == 0, "PaymentSplitter: account already has shares");

        _payees.push(account);
        _shares[account] = shares_;
        _totalShares = _totalShares + shares_;
        emit PayeeAdded(account, shares_);
    }
}

// File: @openzeppelin/contracts/token/ERC20/extensions/draft-IERC20Permit.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts v4.4.1 (token/ERC20/extensions/draft-IERC20Permit.sol)

pragma solidity ^0.8.0;

/**
 * @dev Interface of the ERC20 Permit extension allowing approvals to be made via signatures, as defined in
 * https://eips.ethereum.org/EIPS/eip-2612[EIP-2612].
 *
 * Adds the {permit} method, which can be used to change an account's ERC20 allowance (see {IERC20-allowance}) by
 * presenting a message signed by the account. By not relying on {IERC20-approve}, the token holder account doesn't
 * need to send a transaction, and thus is not required to hold Ether at all.
 */
interface IERC20Permit {
    /**
     * @dev Sets `value` as the allowance of `spender` over ``owner``'s tokens,
     * given ``owner``'s signed approval.
     *
     * IMPORTANT: The same issues {IERC20-approve} has related to transaction
     * ordering also apply here.
     *
     * Emits an {Approval} event.
     *
     * Requirements:
     *
     * - `spender` cannot be the zero address.
     * - `deadline` must be a timestamp in the future.
     * - `v`, `r` and `s` must be a valid `secp256k1` signature from `owner`
     * over the EIP712-formatted function arguments.
     * - the signature must use ``owner``'s current nonce (see {nonces}).
     *
     * For more information on the signature format, see the
     * https://eips.ethereum.org/EIPS/eip-2612#specification[relevant EIP
     * section].
     */
    function permit(
        address owner,
        address spender,
        uint256 value,
        uint256 deadline,
        uint8 v,
        bytes32 r,
        bytes32 s
    ) external;

    /**
     * @dev Returns the current nonce for `owner`. This value must be
     * included whenever a signature is generated for {permit}.
     *
     * Every successful call to {permit} increases ``owner``'s nonce by one. This
     * prevents a signature from being used multiple times.
     */
    function nonces(address owner) external view returns (uint256);

    /**
     * @dev Returns the domain separator used in the encoding of the signature for {permit}, as defined by {EIP712}.
     */
    // solhint-disable-next-line func-name-mixedcase
    function DOMAIN_SEPARATOR() external view returns (bytes32);
}

// File: @openzeppelin/contracts/token/ERC20/IERC20.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts (last updated v4.6.0) (token/ERC20/IERC20.sol)

pragma solidity ^0.8.0;

/**
 * @dev Interface of the ERC20 standard as defined in the EIP.
 */
interface IERC20 {
    /**
     * @dev Emitted when `value` tokens are moved from one account (`from`) to
     * another (`to`).
     *
     * Note that `value` may be zero.
     */
    event Transfer(address indexed from, address indexed to, uint256 value);

    /**
     * @dev Emitted when the allowance of a `spender` for an `owner` is set by
     * a call to {approve}. `value` is the new allowance.
     */
    event Approval(address indexed owner, address indexed spender, uint256 value);

    /**
     * @dev Returns the amount of tokens in existence.
     */
    function totalSupply() external view returns (uint256);

    /**
     * @dev Returns the amount of tokens owned by `account`.
     */
    function balanceOf(address account) external view returns (uint256);

    /**
     * @dev Moves `amount` tokens from the caller's account to `to`.
     *
     * Returns a boolean value indicating whether the operation succeeded.
     *
     * Emits a {Transfer} event.
     */
    function transfer(address to, uint256 amount) external returns (bool);

    /**
     * @dev Returns the remaining number of tokens that `spender` will be
     * allowed to spend on behalf of `owner` through {transferFrom}. This is
     * zero by default.
     *
     * This value changes when {approve} or {transferFrom} are called.
     */
    function allowance(address owner, address spender) external view returns (uint256);

    /**
     * @dev Sets `amount` as the allowance of `spender` over the caller's tokens.
     *
     * Returns a boolean value indicating whether the operation succeeded.
     *
     * IMPORTANT: Beware that changing an allowance with this method brings the risk
     * that someone may use both the old and the new allowance by unfortunate
     * transaction ordering. One possible solution to mitigate this race
     * condition is to first reduce the spender's allowance to 0 and set the
     * desired value afterwards:
     * https://github.com/ethereum/EIPs/issues/20#issuecomment-263524729
     *
     * Emits an {Approval} event.
     */
    function approve(address spender, uint256 amount) external returns (bool);

    /**
     * @dev Moves `amount` tokens from `from` to `to` using the
     * allowance mechanism. `amount` is then deducted from the caller's
     * allowance.
     *
     * Returns a boolean value indicating whether the operation succeeded.
     *
     * Emits a {Transfer} event.
     */
    function transferFrom(
        address from,
        address to,
        uint256 amount
    ) external returns (bool);
}

// File: @openzeppelin/contracts/token/ERC20/utils/SafeERC20.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts (last updated v4.8.0) (token/ERC20/utils/SafeERC20.sol)

pragma solidity ^0.8.0;

import "../IERC20.sol";
import "../extensions/draft-IERC20Permit.sol";
import "../../../utils/Address.sol";

/**
 * @title SafeERC20
 * @dev Wrappers around ERC20 operations that throw on failure (when the token
 * contract returns false). Tokens that return no value (and instead revert or
 * throw on failure) are also supported, non-reverting calls are assumed to be
 * successful.
 * To use this library you can add a `using SafeERC20 for IERC20;` statement to your contract,
 * which allows you to call the safe operations as `token.safeTransfer(...)`, etc.
 */
library SafeERC20 {
    using Address for address;

    function safeTransfer(
        IERC20 token,
        address to,
        uint256 value
    ) internal {
        _callOptionalReturn(token, abi.encodeWithSelector(token.transfer.selector, to, value));
    }

    function safeTransferFrom(
        IERC20 token,
        address from,
        address to,
        uint256 value
    ) internal {
        _callOptionalReturn(token, abi.encodeWithSelector(token.transferFrom.selector, from, to, value));
    }

    /**
     * @dev Deprecated. This function has issues similar to the ones found in
     * {IERC20-approve}, and its usage is discouraged.
     *
     * Whenever possible, use {safeIncreaseAllowance} and
     * {safeDecreaseAllowance} instead.
     */
    function safeApprove(
        IERC20 token,
        address spender,
        uint256 value
    ) internal {
        // safeApprove should only be called when setting an initial allowance,
        // or when resetting it to zero. To increase and decrease it, use
        // 'safeIncreaseAllowance' and 'safeDecreaseAllowance'
        require(
            (value == 0) || (token.allowance(address(this), spender) == 0),
            "SafeERC20: approve from non-zero to non-zero allowance"
        );
        _callOptionalReturn(token, abi.encodeWithSelector(token.approve.selector, spender, value));
    }

    function safeIncreaseAllowance(
        IERC20 token,
        address spender,
        uint256 value
    ) internal {
        uint256 newAllowance = token.allowance(address(this), spender) + value;
        _callOptionalReturn(token, abi.encodeWithSelector(token.approve.selector, spender, newAllowance));
    }

    function safeDecreaseAllowance(
        IERC20 token,
        address spender,
        uint256 value
    ) internal {
        unchecked {
            uint256 oldAllowance = token.allowance(address(this), spender);
            require(oldAllowance >= value, "SafeERC20: decreased allowance below zero");
            uint256 newAllowance = oldAllowance - value;
            _callOptionalReturn(token, abi.encodeWithSelector(token.approve.selector, spender, newAllowance));
        }
    }

    function safePermit(
        IERC20Permit token,
        address owner,
        address spender,
        uint256 value,
        uint256 deadline,
        uint8 v,
        bytes32 r,
        bytes32 s
    ) internal {
        uint256 nonceBefore = token.nonces(owner);
        token.permit(owner, spender, value, deadline, v, r, s);
        uint256 nonceAfter = token.nonces(owner);
        require(nonceAfter == nonceBefore + 1, "SafeERC20: permit did not succeed");
    }

    /**
     * @dev Imitates a Solidity high-level call (i.e. a regular function call to a contract), relaxing the requirement
     * on the return value: the return value is optional (but if data is returned, it must not be false).
     * @param token The token targeted by the call.
     * @param data The call data (encoded using abi.encode or one of its variants).
     */
    function _callOptionalReturn(IERC20 token, bytes memory data) private {
        // We need to perform a low level call here, to bypass Solidity's return data size checking mechanism, since
        // we're implementing it ourselves. We use {Address-functionCall} to perform this call, which verifies that
        // the target address contains contract code and also asserts for success in the low-level call.

        bytes memory returndata = address(token).functionCall(data, "SafeERC20: low-level call failed");
        if (returndata.length > 0) {
            // Return data is optional
            require(abi.decode(returndata, (bool)), "SafeERC20: ERC20 operation did not succeed");
        }
    }
}

// File: @openzeppelin/contracts/utils/Address.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts (last updated v4.8.0) (utils/Address.sol)

pragma solidity ^0.8.1;

/**
 * @dev Collection of functions related to the address type
 */
library Address {
    /**
     * @dev Returns true if `account` is a contract.
     *
     * [IMPORTANT]
     * ====
     * It is unsafe to assume that an address for which this function returns
     * false is an externally-owned account (EOA) and not a contract.
     *
     * Among others, `isContract` will return false for the following
     * types of addresses:
     *
     *  - an externally-owned account
     *  - a contract in construction
     *  - an address where a contract will be created
     *  - an address where a contract lived, but was destroyed
     * ====
     *
     * [IMPORTANT]
     * ====
     * You shouldn't rely on `isContract` to protect against flash loan attacks!
     *
     * Preventing calls from contracts is highly discouraged. It breaks composability, breaks support for smart wallets
     * like Gnosis Safe, and does not provide security since it can be circumvented by calling from a contract
     * constructor.
     * ====
     */
    function isContract(address account) internal view returns (bool) {
        // This method relies on extcodesize/address.code.length, which returns 0
        // for contracts in construction, since the code is only stored at the end
        // of the constructor execution.

        return account.code.length > 0;
    }

    /**
     * @dev Replacement for Solidity's `transfer`: sends `amount` wei to
     * `recipient`, forwarding all available gas and reverting on errors.
     *
     * https://eips.ethereum.org/EIPS/eip-1884[EIP1884] increases the gas cost
     * of certain opcodes, possibly making contracts go over the 2300 gas limit
     * imposed by `transfer`, making them unable to receive funds via
     * `transfer`. {sendValue} removes this limitation.
     *
     * https://diligence.consensys.net/posts/2019/09/stop-using-soliditys-transfer-now/[Learn more].
     *
     * IMPORTANT: because control is transferred to `recipient`, care must be
     * taken to not create reentrancy vulnerabilities. Consider using
     * {ReentrancyGuard} or the
     * https://solidity.readthedocs.io/en/v0.5.11/security-considerations.html#use-the-checks-effects-interactions-pattern[checks-effects-interactions pattern].
     */
    function sendValue(address payable recipient, uint256 amount) internal {
        require(address(this).balance >= amount, "Address: insufficient balance");

        (bool success, ) = recipient.call{value: amount}("");
        require(success, "Address: unable to send value, recipient may have reverted");
    }

    /**
     * @dev Performs a Solidity function call using a low level `call`. A
     * plain `call` is an unsafe replacement for a function call: use this
     * function instead.
     *
     * If `target` reverts with a revert reason, it is bubbled up by this
     * function (like regular Solidity function calls).
     *
     * Returns the raw returned data. To convert to the expected return value,
     * use https://solidity.readthedocs.io/en/latest/units-and-global-variables.html?highlight=abi.decode#abi-encoding-and-decoding-functions[`abi.decode`].
     *
     * Requirements:
     *
     * - `target` must be a contract.
     * - calling `target` with `data` must not revert.
     *
     * _Available since v3.1._
     */
    function functionCall(address target, bytes memory data) internal returns (bytes memory) {
        return functionCallWithValue(target, data, 0, "Address: low-level call failed");
    }

    /**
     * @dev Same as {xref-Address-functionCall-address-bytes-}[`functionCall`], but with
     * `errorMessage` as a fallback revert reason when `target` reverts.
     *
     * _Available since v3.1._
     */
    function functionCall(
        address target,
        bytes memory data,
        string memory errorMessage
    ) internal returns (bytes memory) {
        return functionCallWithValue(target, data, 0, errorMessage);
    }

    /**
     * @dev Same as {xref-Address-functionCall-address-bytes-}[`functionCall`],
     * but also transferring `value` wei to `target`.
     *
     * Requirements:
     *
     * - the calling contract must have an ETH balance of at least `value`.
     * - the called Solidity function must be `payable`.
     *
     * _Available since v3.1._
     */
    function functionCallWithValue(
        address target,
        bytes memory data,
        uint256 value
    ) internal returns (bytes memory) {
        return functionCallWithValue(target, data, value, "Address: low-level call with value failed");
    }

    /**
     * @dev Same as {xref-Address-functionCallWithValue-address-bytes-uint256-}[`functionCallWithValue`], but
     * with `errorMessage` as a fallback revert reason when `target` reverts.
     *
     * _Available since v3.1._
     */
    function functionCallWithValue(
        address target,
        bytes memory data,
        uint256 value,
        string memory errorMessage
    ) internal returns (bytes memory) {
        require(address(this).balance >= value, "Address: insufficient balance for call");
        (bool success, bytes memory returndata) = target.call{value: value}(data);
        return verifyCallResultFromTarget(target, success, returndata, errorMessage);
    }

    /**
     * @dev Same as {xref-Address-functionCall-address-bytes-}[`functionCall`],
     * but performing a static call.
     *
     * _Available since v3.3._
     */
    function functionStaticCall(address target, bytes memory data) internal view returns (bytes memory) {
        return functionStaticCall(target, data, "Address: low-level static call failed");
    }

    /**
     * @dev Same as {xref-Address-functionCall-address-bytes-string-}[`functionCall`],
     * but performing a static call.
     *
     * _Available since v3.3._
     */
    function functionStaticCall(
        address target,
        bytes memory data,
        string memory errorMessage
    ) internal view returns (bytes memory) {
        (bool success, bytes memory returndata) = target.staticcall(data);
        return verifyCallResultFromTarget(target, success, returndata, errorMessage);
    }

    /**
     * @dev Same as {xref-Address-functionCall-address-bytes-}[`functionCall`],
     * but performing a delegate call.
     *
     * _Available since v3.4._
     */
    function functionDelegateCall(address target, bytes memory data) internal returns (bytes memory) {
        return functionDelegateCall(target, data, "Address: low-level delegate call failed");
    }

    /**
     * @dev Same as {xref-Address-functionCall-address-bytes-string-}[`functionCall`],
     * but performing a delegate call.
     *
     * _Available since v3.4._
     */
    function functionDelegateCall(
        address target,
        bytes memory data,
        string memory errorMessage
    ) internal returns (bytes memory) {
        (bool success, bytes memory returndata) = target.delegatecall(data);
        return verifyCallResultFromTarget(target, success, returndata, errorMessage);
    }

    /**
     * @dev Tool to verify that a low level call to smart-contract was successful, and revert (either by bubbling
     * the revert reason or using the provided one) in case of unsuccessful call or if target was not a contract.
     *
     * _Available since v4.8._
     */
    function verifyCallResultFromTarget(
        address target,
        bool success,
        bytes memory returndata,
        string memory errorMessage
    ) internal view returns (bytes memory) {
        if (success) {
            if (returndata.length == 0) {
                // only check isContract if the call was successful and the return data is empty
                // otherwise we already know that it was a contract
                require(isContract(target), "Address: call to non-contract");
            }
            return returndata;
        } else {
            _revert(returndata, errorMessage);
        }
    }

    /**
     * @dev Tool to verify that a low level call was successful, and revert if it wasn't, either by bubbling the
     * revert reason or using the provided one.
     *
     * _Available since v4.3._
     */
    function verifyCallResult(
        bool success,
        bytes memory returndata,
        string memory errorMessage
    ) internal pure returns (bytes memory) {
        if (success) {
            return returndata;
        } else {
            _revert(returndata, errorMessage);
        }
    }

    function _revert(bytes memory returndata, string memory errorMessage) private pure {
        // Look for revert reason and bubble it up if present
        if (returndata.length > 0) {
            // The easiest way to bubble the revert reason is using memory via assembly
            /// @solidity memory-safe-assembly
            assembly {
                let returndata_size := mload(returndata)
                revert(add(32, returndata), returndata_size)
            }
        } else {
            revert(errorMessage);
        }
    }
}

// File: @openzeppelin/contracts/utils/Context.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts v4.4.1 (utils/Context.sol)

pragma solidity ^0.8.0;

/**
 * @dev Provides information about the current execution context, including the
 * sender of the transaction and its data. While these are generally available
 * via msg.sender and msg.data, they should not be accessed in such a direct
 * manner, since when dealing with meta-transactions the account sending and
 * paying for execution may not be the actual sender (as far as an application
 * is concerned).
 *
 * This contract is only required for intermediate, library-like contracts.
 */
abstract contract Context {
    function _msgSender() internal view virtual returns (address) {
        return msg.sender;
    }

    function _msgData() internal view virtual returns (bytes calldata) {
        return msg.data;
    }
}

// File: contracts/BlueSubs.sol



// SPDX-License-Identifier: MIT
pragma solidity ^0.8.0;

import "@openzeppelin/contracts/access/Ownable.sol";
import "@openzeppelin/contracts/finance/PaymentSplitter.sol";

/*
 _____ _         _____     _       
| __  | |_ _ ___|   __|_ _| |_ ___ 
| __ -| | | | -_|__   | | | . |_ -|
|_____|_|___|___|_____|___|___|___|
                                                                
*/

/// @title BlueSubs NFT contract.
contract BlueSubs is Ownable, PaymentSplitter {

    enum Version {
        None,
        LightTracker,
        CompleteTracker,
        ToolsTracker,
        Full
    }

    struct Subscription {
        string discord_id;
        uint date_of_purchase;
        uint8 month;
        Version version;
    }

    struct Receipt {
        address payer;
        uint8 month_added;
        bool is_community;
        string creator_code_used;
    }

    struct Creator {
        address creator;
        uint reduction_percentage;
        uint cashback_percentage;
        string code;
    }

    // Errors
    error MonthlyPlanError();
    error InvalidCreatorCodeError();
    error NotEnoughFundError();
    error AlreadyActivePlanError();
    error CallerIsContractError();
    error CreatorCodeExistError();
    error InvalidPercentageError();
    error InvalidPlanError();
    error InvalidLengthError();
    error NotValidOwnerError(address, address);

    // Events
    event SubscriptionRegister(Receipt receipt, Subscription subscription_updated);
    event CreatorRegister(Creator creator_updated);
    event CreatorCashbackProof(Creator creator, uint value);

    // Private
    mapping(Version => uint) private _communityPrices;
    mapping(Version => uint) private _individualPrices;
    mapping(string => Subscription) private _subscribers;
    mapping(string => Subscription) private _communities;
    mapping(string => Creator) private _creators;
    uint private immutable _teamLength;

    //Constructor of the collection
    constructor(address[] memory _team, uint[] memory _teamShares) PaymentSplitter(_team, _teamShares) {
        _teamLength = _team.length;
        _communityPrices[Version.LightTracker] = 0.04 ether;
        _communityPrices[Version.CompleteTracker] = 0.1 ether;
        _communityPrices[Version.ToolsTracker] = 0.1 ether;
        _communityPrices[Version.Full] = 0.2 ether;

        _individualPrices[Version.LightTracker] = 0.01 ether;
        _individualPrices[Version.CompleteTracker] = 0.02 ether;
        _individualPrices[Version.Full] = 0.03 ether;
    }

    /**
    * @notice Subscribe for a certain a month to an individual version
    * @param discordUserId The discord user's id covered by the subscription
    * @param version The version the user will subscribe too
    * @param month The amount of month for the subscription plan
    * @param code The creator code that will be applied for a reduction bonus
    **/
    function individualSubscribe(string calldata discordUserId, Version version, uint8 month, string calldata code) external payable {
        if(tx.origin != msg.sender) revert CallerIsContractError();
        if(version == Version.None || version == Version.ToolsTracker) revert InvalidPlanError();
        if(month < 1) revert MonthlyPlanError();

        uint versionPrice = _individualPrices[version];
        bool has_code = _creators[code].reduction_percentage != 0;
        if(has_code) {
            uint reduction_percentage = _creators[code].reduction_percentage;
            if(reduction_percentage == 0) revert InvalidCreatorCodeError();
            if(msg.value != (versionPrice - (versionPrice * reduction_percentage / 100 )) * month) revert NotEnoughFundError();
        }
        else {
            if(msg.value != versionPrice * month) revert NotEnoughFundError();
        }

        uint date_of_purchase = _subscribers[discordUserId].date_of_purchase;
        uint8 month_bought = _subscribers[discordUserId].month;
        uint current_time = block.timestamp;
        if(date_of_purchase + (month_bought * 30 days) > current_time) // Previous subscription is still valid
        {
            if(uint(_subscribers[discordUserId].version) != uint(version)) revert AlreadyActivePlanError();
            _subscribers[discordUserId].month += month;
        }
        else 
        {
            _subscribers[discordUserId].discord_id = discordUserId;
            _subscribers[discordUserId].date_of_purchase = current_time;
            _subscribers[discordUserId].month = month;
            _subscribers[discordUserId].version = version;
        }
        emit SubscriptionRegister(Receipt(msg.sender, month, false, code), _subscribers[discordUserId]);

        if(has_code) {
            uint cashback_value = (msg.value * _creators[code].cashback_percentage) / 100;
            emit CreatorCashbackProof(_creators[code], cashback_value);
            payable(_creators[code].creator).transfer(cashback_value);
        }
    }

    /**
    * @notice Subscribe for a certain a month to the community version
    * @param discordGuildId The discord guild's id covered by the subscription
    * @param version The version the community will subscribe too
    * @param month The amount of month for the subscription plan
    **/
    function communitySubscribe(string calldata discordGuildId, Version version, uint8 month) external payable {
        if(tx.origin != msg.sender) revert CallerIsContractError();
        if(version == Version.None) revert InvalidPlanError();
        if(month < 1) revert MonthlyPlanError();
        uint versionPrice = _communityPrices[version];
        if(msg.value != versionPrice * month) revert NotEnoughFundError();

        uint date_of_purchase = _communities[discordGuildId].date_of_purchase;
        uint month_bought = _communities[discordGuildId].month;
        uint current_time = block.timestamp;
        if(date_of_purchase + (month_bought * 30 days) > current_time) // Previous subscription is still valid
        {
            if(uint(_communities[discordGuildId].version) != uint(version)) revert AlreadyActivePlanError();
            _communities[discordGuildId].month += month;
        }
        else 
        {
            _communities[discordGuildId].discord_id = discordGuildId;
            _communities[discordGuildId].date_of_purchase = current_time;
            _communities[discordGuildId].month = month;
            _communities[discordGuildId].version = version;
        }
        emit SubscriptionRegister(Receipt(msg.sender, month, true, ""), _communities[discordGuildId]);
    }

    /**
    * @notice Add an subscriber
    *
    * @param discordUserId The discord user id that will be added
    * @param version The version the user will subscribe too
    * @param month The amount of month for the subscription plan
    */
    function forceSubscriber(string calldata discordUserId, Version version, uint8 month) external {
        _enforceAdmin();
        _subscribers[discordUserId].discord_id = discordUserId;
        _subscribers[discordUserId].date_of_purchase = block.timestamp;
        _subscribers[discordUserId].month = month;
        _subscribers[discordUserId].version = version;
        emit SubscriptionRegister(Receipt(msg.sender, month, false, ""), _subscribers[discordUserId]);
    }

    /**
    * @notice Remove an existing subscriber
    *
    * @param discordUserId The discord user id that will be removed
    */
    function removeExistingSubscriber(string calldata discordUserId) external {
        _enforceAdmin();
        delete _subscribers[discordUserId];
    }

    /**
    * @notice Add a community
    *
    * @param discordGuildId The discord guild id that will be added
    * @param version The version the user will subscribe too
    * @param month The amount of month for the subscription plan
    */
    function forceCommunity(string calldata discordGuildId, Version version, uint8 month) external {
        _enforceAdmin();
        _communities[discordGuildId].discord_id = discordGuildId;
        _communities[discordGuildId].date_of_purchase = block.timestamp;
        _communities[discordGuildId].month = month;
        _communities[discordGuildId].version = version;
        emit SubscriptionRegister(Receipt(msg.sender, month, true, ""), _communities[discordGuildId]);
    }

    /**
    * @notice Remove an existing community
    *
    * @param discordGuildId The discord guild id that will be removed
    */
    function removeExistingCommunity(string calldata discordGuildId) external {
        _enforceAdmin();
        delete _communities[discordGuildId];
    }

    /**
    * @notice Insert a new creator
    *
    * @param creatorAddresses The address of the creators that will be added
    * @param creatorReductionPercentages The reduction percentage that will be applied for all creators
    * @param creatorCashbackPercentages The cashback percentage that will be applied for all creators
    * @param creatorCodes The creator codes that will need to be used to retrieve the percentage
    */
    function addNewCreators(address[] calldata creatorAddresses, uint[] calldata creatorReductionPercentages, uint[] calldata creatorCashbackPercentages, string[] calldata creatorCodes, uint totalQuantity) external {
        _enforceAdmin();
        uint testLength = creatorAddresses.length;
        if(testLength != totalQuantity) revert InvalidLengthError();
        testLength = creatorReductionPercentages.length;
        if(testLength != totalQuantity) revert InvalidLengthError();
        testLength = creatorCashbackPercentages.length;
        if(testLength != totalQuantity) revert InvalidLengthError();
        testLength = creatorCodes.length;
        if(testLength != totalQuantity) revert InvalidLengthError();
        
        for(uint i = 0; i < totalQuantity; i++)
        {
            uint creatorReductionPercentage = creatorReductionPercentages[i];
            uint creatorCashbackPercentage = creatorCashbackPercentages[i];
            if(creatorReductionPercentage > 100 || creatorCashbackPercentage > 100 || creatorReductionPercentage == 0 || creatorCashbackPercentage == 0) revert InvalidPercentageError();
            string calldata creatorCode = creatorCodes[i];
            if(_creators[creatorCode].reduction_percentage != 0) revert CreatorCodeExistError();
            _creators[creatorCode].creator = creatorAddresses[i];
            _creators[creatorCode].reduction_percentage = creatorReductionPercentage;
            _creators[creatorCode].cashback_percentage = creatorCashbackPercentage;
            _creators[creatorCode].code = creatorCode;
            emit CreatorRegister(_creators[creatorCode]);
        }
    }


    /**
    * @notice Update the wallet of an existing creator
    *
    * @param creatorCode The creator code that will be removed
    * @param newWallet The new address for the creator
    */
    function updateExistingCreator(string calldata creatorCode, address newWallet) external {
        _enforceAdmin();
        if(_creators[creatorCode].reduction_percentage == 0) revert InvalidCreatorCodeError();
        _creators[creatorCode].creator = newWallet;
    }

    /**
    * @notice Remove an existing creator
    *
    * @param creatorCode The creator code that will be removed
    */
    function removeExistingCreator(string calldata creatorCode) external {
        _enforceAdmin();
        delete _creators[creatorCode];
    }

    /**
    * @notice Change the price of an individual specified version
    *
    * @param version The version whose price will change
    * @param newPrice The new price for the version
    */
    function setIndividualPrice(Version version, uint newPrice) external {
        _enforceAdmin();
        _individualPrices[version] = newPrice;
    }

    /**
    * @notice Change the price of a community specified version
    *
    * @param version The version whose price will change
    * @param newPrice The new price for the version
    */
    function setCommunityPrice(Version version, uint newPrice) external {
        _enforceAdmin();
        _communityPrices[version] = newPrice;
    }

    /**
    * @notice Pay everyone in the team
    */
    function releaseAll() external {
        _enforceAdmin();
        for(uint i = 0 ; i < _teamLength ; i++) {
            release(payable(payee(i)));
        }
    }

    /**
    * @notice Return the individual price of a specified version
    *
    * @param version The version to get the price of
    *
    * @return price The price of the version
    **/
    function getIndividualPrice(Version version) public view returns (uint price) {
        return _individualPrices[version];
    }

    /**
    * @notice Return the community price of a specified version
    *
    * @param version The version to get the price of
    *
    * @return price The price of the version
    **/
    function getCommunityPrice(Version version) public view returns (uint price) {
        return _communityPrices[version];
    }

    /**
    * @notice Return a creator associated to the creator code
    *
    * @param creatorCode The creator code to search for
    *
    * @return creator The creator associated to the code
    **/
    function getCreator(string memory creatorCode) public view returns (Creator memory creator) {
        return _creators[creatorCode];
    }

    /**
    * @notice Return a subscriber associated to a discord user id
    *    
    * @param discordUserId The discord user id to search for
    *
    * @return subscriber The subscriber associated to the discord user id
    **/
    function getSubscriber(string calldata discordUserId) public view returns (Subscription memory subscriber) {
        return _subscribers[discordUserId];
    }

    /**
    * @notice Return a community associated to a discord guild id
    *    
    * @param discordGuildId The discord guild id to search for
    *
    * @return community The community associated to the discord guild id
    **/
    function getCommunity(string calldata discordGuildId) public view returns (Subscription memory community) {
        return _communities[discordGuildId];
    }

    function _enforceAdmin() internal view {
        if (owner() != msg.sender) revert NotValidOwnerError(address(this), owner());
    }

    receive() override external payable {
        revert('Cannot send money to contract');
    }
}